

ME 201/MTH 281/ME400/CHE400

Bessel Functions

1. Introduction

This notebook has two purposes: to give a brief introduction to Bessel functions, and to illustrate how *Mathematica* can be used in working with Bessel functions. We begin with a summary of the origin of Bessel's equation in our course.

By separating variables for the Laplace equation in cylindrical coordinates, we found solutions of the form

$$F(r)G(z)\begin{pmatrix} \cos(m\theta) \\ \sin(m\theta) \end{pmatrix},$$

where $m = 0, 1, 2, \dots$. The radial function F satisfied Bessel's equation of order m with a parameter λ :

$$\frac{d}{dr} \left(r \frac{dF}{dr} \right) + \left(\lambda r - \frac{m^2}{r} \right) F = 0. \quad (1)$$

Here λ is a constant arising in the separation process. By introducing a new independent variable

$$x = \sqrt{\lambda} r,$$

we may put equation (1) in the form

$$\frac{d}{dx} \left(x \frac{dF}{dx} \right) + \left(x - \frac{m^2}{x} \right) F = 0. \quad (2)$$

We deal with the solutions of this equation for general m in section 3.

The important special case $m = 0$ corresponds to solutions of the Laplace equation which are independent of θ . The equation for F then reduces to Bessel's equation of order 0:

$$\frac{d}{dx} \left(x \frac{dF}{dx} \right) + xF = 0. \quad (3)$$

We deal with solutions of this equation in section 2.

The equations (2) and (3) arise in those cases where the solution is oscillatory in r and exponential in z . In other problems, the homogeneous boundary conditions are on surfaces of constant z , making the solutions oscillatory in z . In those cases, the separation constant has a sign opposite to that in equation (1), and the resulting radial equation has the form

$$\frac{d}{dr} \left(r \frac{dF}{dr} \right) - \left(\lambda r + \frac{m^2}{r} \right) F = 0. \quad (4)$$

This is called the modified Bessel's equation of order m with a parameter λ . The scaling $x = \sqrt{\lambda} r$ reduces this to

$$\frac{d}{dx} \left(x \frac{dF}{dx} \right) - \left(x + \frac{m^2}{x} \right) F = 0. \quad (5)$$

Solutions of this equation are discussed in section 5. The special case $m = 0$ for axisymmetric solutions is

$$\frac{d}{dx} \left(x \frac{dF}{dx} \right) - xF = 0 . \quad (6)$$

and solutions of this equation are discussed in section 4.

2. The Bessel Functions J_0 and Y_0

■ 2.1 Summary of Solution by the Method of Frobenius

As shown in class, equation (3) has a regular singular point at $x = 0$, and therefore can be solved by the method of Frobenius. The detailed calculations show that the indicial equation has a repeated root 0, 0. Thus there is only one solution of the Frobenius form, and the second solution has a logarithmic singularity at $x = 0$. Because the indicial root is 0, the Frobenius solution has the form of a power series and is thus well-behaved at $x = 0$. This solution has been standardized by choosing the multiplicative constant so that the function value is 1 when $x = 0$. The resulting function is called the Bessel function of the first kind of order 0, and is denoted by J_0 . The second solution -- the one with a logarithmic singularity at $x = 0$ -- has also been standardized, and it is denoted by Y_0 .

■ 2.2 Series Expansions for J_0 and Y_0

The Frobenius method readily yields the following series for J_0 :

$$J_0(x) = 1 - \frac{(x/2)^2}{(1!)^2} + \frac{(x/2)^4}{(2!)^2} - \frac{(x/2)^6}{(3!)^2} + \dots = \sum_{k=0}^{\infty} \frac{(-1)^k (x/2)^{2k}}{(k!)^2} . \quad (7)$$

Because the equation has no other singular points, we expect this series to converge for all x , a conclusion easily verified by the ratio test.

The derivation of a series for $Y_0(x)$ is much more difficult, and the resulting series is more complicated:

$$Y_0(x) = \frac{2}{\pi} \left\{ \ln\left(\frac{1}{2}x\right) + \gamma \right\} J_0(x) + \frac{2}{\pi} \left\{ \frac{(x/2)^2}{(1!)^2} - \left(1 + \frac{1}{2}\right) \frac{(x/2)^4}{(2!)^2} + \left(1 + \frac{1}{2} + \frac{1}{3}\right) \frac{(x/2)^6}{(3!)^2} - \dots \right\} \quad (8)$$

Here γ is Euler's constant which is known to *Mathematica* by the name EulerGamma:

```
N[EulerGamma]
```

```
0.577216
```

Fortunately for us, we do not have to use these series directly to get values for these functions. Bessel functions are built-in to *Mathematica*, and we consider that next.

■ 2.3 Using *Mathematica* to Evaluate J_0 and Y_0

The *Mathematica* function BesselJ[m,x] returns the value of $J_m(x)$ and the function BesselY[m,x] returns the value of $Y_m(x)$. As an experiment, let's calculate the value of J_0 for $x = 1, 2$, and 3, using both the series of equation (7) and the *Mathematica* function. We first define the Nth partial sum of the series.

```
Jseries[x_, n_] := Sum[(-1)^k * (x / 2)^(2 * k) / (k!)^2, {k, 0, n}]
```

We start with $x = 1$. *Mathematica* gives

```
BesselJ[0, 1.0]
```

```
0.765198
```

The series gives

```
Table[Jseries[1.0, n], {n, 1, 4}]
```

```
{0.75, 0.765625, 0.765191, 0.765198}
```

Thus four terms are sufficient to give six-place accuracy. For $x = 2$ and 3 , we expect that we will need more terms in the series.

```
BesselJ[0, 2.0]
```

```
0.223891
```

```
Table[Jseries[2.0, n], {n, 4, 6}]
```

```
{0.223958, 0.223889, 0.223891}
```

For $x = 2$, we see that 6 terms are sufficient for six-place accuracy.

```
BesselJ[0, 3.0]
```

```
-0.260052
```

```
Table[Jseries[3.0, n], {n, 5, 7}]
```

```
{-0.260291, -0.260041, -0.260052}
```

For $x = 3$, we need 7 terms.

From here on, we will use the *Mathematica* built-in functions because they are much more convenient. In any case, the series cannot be used for very large values of x , something which *Mathematica* knows to avoid.


At this point we have considerable machinery, but we have no intuitive feeling for Bessel functions. We remedy that by looking at some graphs.

■ 2.4 Plots of J_0 and Y_0

An appropriate size for printed graphs is 250. For display of graphs, a size of 350 is better. The size is controlled throughout the notebook by the `SetOption` command below. After executing that, we use the function `Plot` to get some pictures of J_0 and Y_0 . First we plot J_0 .

```
SetOptions[Plot, ImageSize -> 250];
```


```
graphJ0 = Plot[BesselJ[0, x], {x, 0, 20}, AxesLabel -> {"x", "J0"}]
```



An interesting graph. We see that J_0 is an oscillatory function. In fact it looks a lot like a damped trig function, an observation which we will make more precise in section 2.6. We also see that J_0 has many zeros (in fact it has infinitely many). As we have already seen in class, we need to find these zeros to find the eigenvalues for solutions of Laplace's equation by separation of variables. In section 2.5, we will see how to do this.


Now let's plot Y_0 . We will have to be a little careful because of the singularity at $x = 0$, so we start our plot a little away from 0. Let's use the same plot range that *Mathematica* selected above for J_0 .

```
graphY0 =
  Plot[BesselY[0, x], {x, 0.01, 20}, PlotRange -> {-0.4, 1.0}, AxesLabel -> {"x", "Y0"}]
```



The function Y_0 , like J_0 , is oscillatory, and again looks like a damped trig function, apart from the divergence to $-\infty$ at $x = 0$. Let's plot the two functions on the same graph:

```
Show[graphJ0, graphY0]
```



Now we see that for larger x they look like damped trig functions differing only by a phase shift. We will see in section 2.6 the truth of that statement.

■ 2.5 Finding Roots of J_0

Earlier in our course, we have seen examples of finding roots by using FindRoot. That will also work for the Bessel functions we are studying here. For J_0 , for example, we see from the graph that the first root is between 2 and 3. Thus we can use FindRoot with an initial guess of 2.5:

```
root1 = FindRoot[BesselJ[0, x], {x, 2.5}]
```

```
{x -> 2.40483}
```

The answer is in the form of a replacement rule. Let's check it by evaluating J_0 at that value:

```
BesselJ[0, x] /. root1
```

```
 $-9.51574 \times 10^{-17}$ 
```

That's close enough to zero for most purposes. We could continue finding the other roots of J_0 by altering the initial guess and repeating the calculation. Fortunately, there is a much easier way in *Mathematica*. *Mathematica* has built-in functions for getting the zeros of Bessel functions. For the J Bessel function the name of the function returning a zero is BesselJZero[n,k]. This returns the kth positive zero of J_n . For convenience we use this function to construct a list (Table) of the first 40 zeros of J_0 . We assign the list to zerolist.

```
zerolist = N[Table[BesselJZero[0, i], {i, 1, 40}]]
```

```
{2.40483, 5.52008, 8.65373, 11.7915, 14.9309, 18.0711, 21.2116, 24.3525, 27.4935, 30.6346,
 33.7758, 36.9171, 40.0584, 43.1998, 46.3412, 49.4826, 52.6241, 55.7655, 58.907, 62.0485,
 65.19, 68.3315, 71.473, 74.6145, 77.756, 80.8976, 84.0391, 87.1806, 90.3222, 93.4637,
 96.6053, 99.7468, 102.888, 106.03, 109.171, 112.313, 115.455, 118.596, 121.738, 124.879}
```

The zeros are then available for any subsequent calculation. For example, the first zero is given by

```
root1 = zerolist[[1]]
```

```
2.40483
```

This is the same as the value found by FindRoot. As before, we check this by evaluating J_0 for this value:

```
BesselJ[0, x] /. x -> root1
```

```
 $-9.51574 \times 10^{-17}$ 
```

We can learn something interesting about the zeros by calculating the spacing between successive zeros:

```
Table[(zerolist[[n + 1]] - zerolist[[n]]), {n, 1, 39}]
```

```
{3.11525, 3.13365, 3.13781, 3.13938, 3.14015, 3.14057, 3.14083, 3.14101, 3.14113, 3.14121,
 3.14128, 3.14133, 3.14137, 3.1414, 3.14142, 3.14144, 3.14146, 3.14147, 3.14149, 3.1415,
 3.1415, 3.14151, 3.14152, 3.14152, 3.14153, 3.14153, 3.14154, 3.14154, 3.14155, 3.14155,
 3.14155, 3.14155, 3.14156, 3.14156, 3.14156, 3.14156, 3.14156, 3.14156, 3.14157, 3.14157}
```

We see that although the zeros are not exactly equally spaced, they are apparently asymptotically equally spaced, and the spacing looks suspiciously like π ! More on this later.

The main point in this section is that the roots of Bessel functions are easily and instantly available.

■ 2.6 Asymptotic Values of J_0 and Y_0 for Large Argument

As we saw earlier, the graphs of the Bessel functions look a lot like lightly damped trig functions. By advanced techniques of asymptotic analysis, it can be shown that for large x , $J_0(x)$ and $Y_0(x)$ are approximated by $J0Large[x]$ and $Y0Large[x]$, where


```
J0Large[x_] := Sqrt[2.0 / (π * x)] * Cos[x - π / 4]
```

```
Y0Large[x_] := Sqrt[2.0 / (π * x)] * Sin[x - π / 4]
```

We see that these are indeed damped trig functions for large x , but the amplitude decrease is very gradual. It goes like $1/\sqrt{x}$, which is much more gradual than exponential damping. We also see that for large x , J_0 and Y_0 differ only by a phase shift. Finally, we see that the larger zeros of J_0 are the zeros of $\cos[x - \pi/4]$ -- that is $(n - 1/4)\pi$ -- and this explains our earlier observation that the interval between zeros appears to approach π .

We compare the exact and asymptotic functions graphically, showing the exact as a solid curve, and the asymptotic as a dashed curve.

```
compJgraph = Plot[{BesselJ[0, x], J0Large[x]}, {x, 0, 10},
  PlotRange -> {-0.41, 1.}, PlotStyle -> {Dashing[{0.01, 0.}], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "J0(x)"}, PlotLabel -> "Exact (solid)
  Asymptotic (dashed)"]
```



We see that the approximation is quite good, especially for x about 3 or larger. Let's look at the numbers:

```
BesselJ[0, 3.0]
```

```
-0.260052
```


J0Large[3.0]

-0.276507

There is roughly a 5% difference.

Finally, we compare graphically the exact and asymptotic forms for Y_0 :

```
compYgraph = Plot[{BesselY[0, x], Y0Large[x]}, {x, 0.01, 10},
  PlotRange -> {-0.4, 1.}, PlotStyle -> {Dashing[{0.01, 0.}], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "Y0(x)"}, PlotLabel -> "Exact (solid)
  Asymptotic (dashed)"]
```



Again excellent agreement for x greater than about 3. Here are the numbers for $x = 3$:

BesselY[0, 3.0]

0.37685

Y0Large[3.0]

0.368443

Only about a 2% difference.

3. The Bessel Functions J_m and Y_m

■ 3.1 Summary of Solution by the Method of Frobenius

The functions $J_m(x)$ and $Y_m(x)$ are the solutions of equation (2), which is Bessel's equation of order m . As shown in class, equation (2) has a regular singular point at $x = 0$, and therefore can be solved by the method of Frobenius. The detailed calculations show that the indicial equation has roots $-m, m$. These differ by an integer, and so the Frobenius Theorem tells us that we will get a solution of the Frobenius form corresponding to the root $+m$, and that there is the possibility of a logarithmic singularity at $x = 0$ for the other solution. Detailed analysis confirms that the second solution is not of the Frobenius form and does indeed have a logarithmic singularity at $x = 0$. The Frobenius solution is of the form of a power series times x^m and is thus well-behaved at $x = 0$ (under the assumption made throughout this notebook that m is a non-negative integer). This well-behaved solution, when standardized as in section 3.2, is called the Bessel function of the first kind of order m , and is denoted by $J_m(x)$. The singular solution, standardized in section 3.2, is called the Bessel function of the second kind of order m , and is denoted by $Y_m(x)$.

■ 3.2 Series Expansions for J_m and Y_m

The Frobenius method readily yields a series for the solution associated with the indicial root $+m$. With the multiplicative constant chosen in a standard way, this series becomes the function J_m :

$$J_m(x) = \left(\frac{x}{2}\right)^m \left(\frac{1}{m!} - \frac{(x/2)^2}{(1!)(1+m)!} + \frac{(x/2)^4}{(2!)(2+m)!} - \frac{(x/2)^6}{(3!)(3+m)!} + \dots \right)$$

(9)

$$= \left(\frac{x}{2}\right)^m \sum_{k=0}^{\infty} \frac{(-1)^k (x/2)^{2k}}{(k!) (k+m)!}.$$

Because the equation has no other singular points, the series converges for all x , a conclusion easily verified by the ratio test.

The derivation of a series for $Y_m(x)$ is much more difficult, and the resulting series is complicated:

$$Y_m(x) = -\frac{(x/2)^{-m}}{\pi} \sum_{k=0}^{m-1} \frac{(m-k-1)!}{k!} \left(\frac{x}{2}\right)^{2k} + \frac{2}{\pi} \ln\left(\frac{1}{2}x\right) J_m(x) - \frac{(x/2)^m}{\pi} \sum_{k=0}^{\infty} \{\psi(k+1) + \psi(m+k+1)\} \frac{(-1)^k (x/2)^{2k}}{k! (m+k)!}, \quad (10)$$

where ψ is the Digamma function, given by $\psi(1) = -\gamma$ (γ is Euler's constant as before), and $\psi(n) = -\gamma + \sum_{k=1}^{n-1} 1/k$.

Notice that the logarithm is not the worst singularity in Y_m at $x = 0$. There are negative powers of x also, with the highest being x^{-m} .

We do not have to use these series directly because these functions are built in to *Mathematica*. We consider that next.

■ 3.3 Using *Mathematica* to Evaluate J_m and Y_m

The *Mathematica* function `BesselJ[m,x]` returns the value of $J_m(x)$ and the function `BesselY[m,x]` returns the value of $Y_m(x)$. As an experiment, let's calculate the value of J_m for $x = 2$, and for $m = 1, 2$, and 3 , using both the series of equation (9) and *Mathematica*. We first define the Nth partial sum of the series.

```
Jseries[x_, m_, n_] := (x/2)^m * Sum[(-1)^k * (x/2)^(2*k) / ((k!) * (k+m)!), {k, 0, n}]
```

We start with $m = 1$. *Mathematica* gives

```
BesselJ[1, 2.0]
```

```
0.576725
```

The series gives

```
Table[Jseries[2.0, 1, n], {n, 1, 5}]
```

```
{0.5, 0.583333, 0.576389, 0.576736, 0.576725}
```

Thus five terms are sufficient to give six-place accuracy for $x = 2$. Now we try $m = 2$.

```
BesselJ[2, 2.0]
```

```
0.352834
```

```
Table[Jseries[2.0, 2, n], {n, 3, 6}]
```

```
{0.352778, 0.352836, 0.352834, 0.352834}
```

Thus again five terms are sufficient for six-place accuracy for $x = 2$. Finally, we try $m = 3$.

```
BesselJ[3, 2.0]
```

```
0.128943
```

```
Table[Jseries[2.0, 3, n], {n, 3, 6}]
```

```
{0.128935, 0.128943, 0.128943, 0.128943}
```

Here four terms are sufficient.


From here on, we will use the *Mathematica* built-in functions because they are much more convenient. In any case, the series can be used effectively only for moderately small values of $|x|$.

At this point, we don't have a good picture of how the Bessel functions depend on the order m . We next look at some graphs to illustrate this.

■ 3.4 Plots of J_m and Y_m

To look at the dependence of J_m on the order m , we plot the first six Bessel functions over the x -range $\{0, 20\}$.


```
graphJm = Plot[{BesselJ[0, x], BesselJ[1, x], BesselJ[2, x], BesselJ[3, x], BesselJ[4, x],
  BesselJ[5, x]}, {x, 0, 20}, PlotRange → {-0.41, 1.}, AxesLabel → {"x", "Jm"}]
```



An interesting graph, although a little busy. The graph illustrates the fact that the larger m is, the more slowly the Bessel function starts up. This happens because J_m is proportional to x^m . Thus at $x = 0$, J_m and its first $m - 1$ derivatives vanish. After this initial sluggish start for the larger m 's, we see that the J_m 's have similar behavior -- that is, they all look like damped trig functions (which we look at in more detail in section 3.6) and they all have infinitely many zeros (which we learn how to find in section 3.5).

Now let's plot Y_m . We will have to be a little careful because of the singularity at $x = 0$, so we start our plot a little away from 0. Let's use the same m -values as above for the J 's.

```
graphYm = Plot[{BesselY[0, x], BesselY[1, x], BesselY[2, x], BesselY[3, x], BesselY[4, x],
  BesselY[5, x]}, {x, 0.01, 20}, PlotRange → {-0.41, 1.}, AxesLabel → {"x", "Jm"}]
```



The functions with the higher values of m diverge to infinity more rapidly as $x \rightarrow 0$, because of the x^{-m} term, and this makes it easy to tell which curve is which on the graph. After the initial divergence, the Y 's settle into the by-now familiar damped oscillation. You may have noticed that the above graph took longer to produce than the previous one for the J 's. That's because the computational effort to find values of the Y 's is much greater than that for finding the J 's.

■ 3.5 Finding Roots of J_m

Earlier we found roots of J_0 . In the same way, we can find the roots of any J_m . As an example, let's find the first 40 zeros of J_5 .

```
zerolist = N[Table[BesselJZero[5, i], {i, 1, 40}]]
```

```
{8.77148, 12.3386, 15.7002, 18.9801, 22.2178, 25.4303, 28.6266, 31.8117, 34.9888, 38.1599,
  41.3264, 44.4893, 47.6494, 50.8072, 53.963, 57.1173, 60.2702, 63.4221, 66.5729, 69.7229,
  72.8722, 76.0208, 79.1689, 82.3164, 85.4636, 88.6103, 91.7567, 94.9028, 98.0485, 101.194,
  104.339, 107.484, 110.629, 113.774, 116.918, 120.063, 123.207, 126.351, 129.495, 132.639}
```

The zeros are then available for any subsequent calculation. For example, the first zero is given by


```
root1 = zerolist[[1]]
```

```
8.77148
```

We check this by evaluating J_5 for this value:

```
BesselJ[5, x] /. x -> root1
```

```
 $6.54068 \times 10^{-15}$ 
```

Excellent accuracy. As we did before for J_0 , let's check the spacing between these zeros:

```
Table[(zerolist[[n + 1]] - zerolist[[n]]), {n, 1, 39}]
```

```
{3.56712, 3.36157, 3.27996, 3.23767, 3.21254, 3.19628, 3.1851, 3.17706, 3.17109, 3.16651,
 3.16294, 3.16008, 3.15777, 3.15586, 3.15428, 3.15294, 3.15181, 3.15084, 3.15, 3.14927,
 3.14863, 3.14807, 3.14757, 3.14713, 3.14674, 3.14638, 3.14607, 3.14578, 3.14552, 3.14528,
 3.14506, 3.14487, 3.14469, 3.14452, 3.14437, 3.14422, 3.14409, 3.14397, 3.14386}
```

We see again that although the zeros are not exactly equally spaced, they seem to be approaching equal spacing for the large roots, and as before the spacing seems to be approaching π . More on this later.

From our graph earlier of J_m for $m = 0, 1, 2, 3, 4$, and 5 , we would expect the first root of J_m to increase with m . Let's check this by finding the first root of each of the first 11 Bessel functions:

```
N[Table[BesselJZero[m, 1], {m, 0, 10}]]
```

```
{2.40483, 3.83171, 5.13562, 6.38016, 7.58834,
 8.77148, 9.93611, 11.0864, 12.2251, 13.3543, 14.4755}
```

Our expectation was correct.

As we saw before, the functions `BesselJZero` and `BesselYZero` make the roots of Bessel functions easily and instantly available.

■ 3.6 Asymptotic Values of J_m and Y_m for Large Argument

The asymptotic formulas we quoted earlier for J_0 and Y_0 have extensions to J_m and Y_m . It can be shown that for large x , $J_m(x)$ and $Y_m(x)$ are approximated by `JLarge[m,x]` and `YLarge[m,x]`, where


```
JLarge[m_, x_] := Sqrt[2.0 / (π * x)] * Cos[x - m * π / 2 - π / 4]
```

```
YLarge[m_, x_] := Sqrt[2.0 / (π * x)] * Sin[x - m * π / 2 - π / 4]
```

These formulas include J_0 and Y_0 as special cases. As before, we see that the amplitude decrease is very gradual. It goes like $1/\sqrt{x}$ and is much more gradual than exponential damping. We also see that for large x , J_m and Y_m differ only by a phase shift. Finally, we see that the large zeros of J_m are the zeros of $\cos[x - m\pi/2 - \pi/4]$ -- that is $(n + m/2 - 1/4)\pi$ -- and this explains our earlier observation that the interval between zeros approaches π .

We compare the exact and asymptotic functions graphically, showing the exact as a solid curve, and the asymptotic as a dashed curve. By way of example, we do this for $m = 5$.


```
compJmgraph = Plot[{BesselJ[5, x], JLarge[5, x]}, {x, 0, 40},
  PlotRange -> {-0.41, 1.}, PlotStyle -> {Dashing[{0.01, 0.}], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "J5(x)"}, PlotLabel -> "Exact (solid)
Asymptotic (dashed)"]
```



We see that the approximation eventually joins up with the exact function, but we have to go to quite large x for this to happen. This is in marked contrast with the case for J_0 , where the results were good for x as small as 3. This is typical of the higher order Bessel functions. For larger m it is even worse.

Now we do the same comparison for Y_5 :

```
compY5graph = Plot[{BesselY[5, x], YLarge[5, x]}, {x, 0.01, 40},
  PlotRange -> {-0.4, 1.0}, PlotStyle -> {Dashing[{0.01, 0.0}], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "Y5(x)"}, PlotLabel -> "Exact (solid)
Asymptotic (dashed)"]
```



Again we must go to very large x to get good agreement. For both functions one can derive more accurate multi-term approximations for large x . We will look at examples of such approximations in our discussion of the modified Bessel functions.

■ 3.7 Recurrence Relations

It is an interesting property of Bessel functions that any three functions of consecutive order are related -- in fact there are two independent such relations. They are

$$J_{m-1}(x) + J_{m+1}(x) = \frac{2m}{x} J_m(x) \quad , \quad (11)$$

and

$$J_{m-1}(x) - J_{m+1}(x) = 2J'_m(x) \quad . \quad (12)$$

The exact same formulas are also valid for $Y_m(x)$ or even for any fixed linear combination of J_m and Y_m . The formulas are valid even for negative indices, provided we use the standard definitions of J_m and Y_m for negative order:

$$J_{-m}(x) = (-1)^m J_m(x) \quad , \quad Y_{-m}(x) = (-1)^m Y_m(x) \quad . \quad (13)$$

The recurrence formulas have surprisingly many uses. For example, a computational scheme for evaluating $J_m(x)$ for fixed x and many different values of m can be designed around equation (11). The formulas are also some-

times used in the evaluation of integrals involving Bessel functions, as we shall see in the next section.

■ 3.8 Integrals Involving Bessel Functions

In completing the solution of the Laplace and other equations by separation of variables in cylindrical coordinates, one usually has to expand some boundary function in a series of Bessel functions. This in turn requires the evaluation of certain integrals involving the Bessel functions. With complicated boundary functions these integrals must be done numerically, which is no problem in *Mathematica* with the `NIntegrate` function. For certain simple cases of boundary functions, the integrals can be done analytically, and these simple cases are often useful as test and calibration cases for the numerical approach. We consider in this section two elementary integrals that can be done analytically.

For the first example, we start with the recurrence formulas (11) and (12). We multiply (11) by x^m and (12) by x^m and add them up. The result can be put into the form

$$\frac{d}{dx} [x^m J_m(x)] = x^m J_{m-1}(x) ,$$

and by integrating this, we get the following indefinite integral:

$$\int x^m J_{m-1}(x) dx = x^m J_m(x) . \quad (14)$$

A frequently occurring special case is given by $m = 1$:

$$\int x J_0(x) dx = x J_1(x) . \quad (15)$$

By integration-by-parts, we can parlay this into other integrals. For example,

$$\int x^3 J_0(x) dx$$

can be evaluated by letting $u = x^2$ and $dv = x J_0(x) dx$, hence $v = x J_1(x)$. The integration-by-parts then gives

$$\int x^3 J_0(x) dx = x^3 J_1(x) - 2 x^2 J_2(x) . \quad (16)$$

Mathematica knows how to do these integrals analytically. Consider the integral given in (15):

```
Integrate[x * BesselJ[0, x], x]
```

$$\frac{1}{2} x^2 \text{Hypergeometric0F1Regularized}\left[2, -\frac{x^2}{4}\right]$$

Not what we expected! Let's see what `FullSimplify` does for us.

```
FullSimplify[%]
```

```
x BesselJ[1, x]
```

That's better.

Here is the *Mathematica* version of the integral in (16):

```
Integrate[x^3 * BesselJ[0, x], x]
```

$$-x^2 (-2 \text{BesselJ}[2, x] + x \text{BesselJ}[3, x])$$

By using equation (11), you can show that this is equivalent to equation (16).

4. The Bessel Functions I_0 and K_0

The functions I_0 and K_0 considered in this section are solutions of equation (6), the modified Bessel's equation of order 0. In principle, all of the development for J_0 and Y_0 could be repeated here with the minor variations required

by the switch from J_0 and Y_0 to I_0 and K_0 . In practice, the functions I_0 and K_0 are not used as much in our course as J_0 and Y_0 , so we will keep the discussion very brief.

The modified Bessel's equation has a regular singular point at $x = 0$. The application of the method of Frobenius shows that the indicial equation has repeated roots $0, 0$, so there is one solution of the Frobenius form, and a second solution with a logarithmic singularity. The solution well-behaved at $x = 0$ is standardized as $I_0(x)$, and the Frobenius series is


$$I_0(x) = 1 + \frac{(x/2)^2}{(1!)^2} + \frac{(x/2)^4}{(2!)^2} + \frac{(x/2)^6}{(3!)^2} + \dots = \sum_{k=0}^{\infty} \frac{(x/2)^{2k}}{(k!)^2}. \quad (17)$$

The series for the second solution K_0 is more difficult to obtain and more intricate. It is given by

$$K_0(x) = -\left\{\ln\left(\frac{1}{2}x\right) + \gamma\right\}I_0(x) + \left\{\frac{(x/2)^2}{(1!)^2} + \left(1 + \frac{1}{2}\right)\frac{(x/2)^4}{(2!)^2} + \left(1 + \frac{1}{2} + \frac{1}{3}\right)\frac{(x/2)^6}{(3!)^2} - \dots\right\}. \quad (18)$$


These functions are both built in. The *Mathematica* function `BesselI[0,x]` gives $I_0(x)$ and the function `BesselK[0,x]` gives $K_0(x)$. Let's see what they look like.

```
graphI0 = Plot[BesselI[0, x], {x, 0, 4},  
PlotRange -> {{0, 4}, {-0.1, 11.5}}, AxesLabel -> {"x", "I0(x)"}]
```



We see that I_0 starts at 1 and appears to increase exponentially. Now we plot K_0 , avoiding the singularity at $x = 0$.

```
graphK0 = Plot[BesselK[0, x], {x, 0.01, 4}, AxesLabel -> {"x", "K0(x)"}]
```



We see that K_0 appears to drop exponentially. As with J_0 and Y_0 , it is possible to derive simple asymptotic approximations for large argument. One can show that for large x , $I_0(x)$ is approximated by `I0Large[x]`, and that $K_0(x)$ is approximated by `K0Large[x]`, where

```
I0Large[x_] := Exp[x] / Sqrt[2 * π * x]
```


```
K0Large[x_] := Exp[-x] * Sqrt[π / (2 * x)]
```

We compare the exact and asymptotic formulas graphically, starting with I_0 .

```

compIgraph = Plot[{BesselI[0, x], IOLarge[x]}, {x, 0.01, 4},
  PlotRange -> {0, 15.0}, PlotStyle -> {Dashing[0], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "I0(x)"}, PlotLabel -> "Exact (solid)
Asymptotic (dashed)"]

```




We see that the approximation is adequate although not spectacular. It is possible to develop more accurate approximations which essentially provide a small correction to $IOLarge$. Let's look at an example of such an approximation here. From Abramowitz and Stegun (p. 377, formula 9.7.1) we get

```

IOLargeMod[x_] := (Exp[x] / Sqrt[2 π x]) (1 + 1 / (8 x))
compIgraph = Plot[{BesselI[0, x], IOLargeMod[x]}, {x, 0.01, 4},
  PlotRange -> {0, 15.0}, PlotStyle -> {Dashing[0], Dashing[{0.01, 0.01}]},
  AxesLabel -> {"x", "I0(x)"}, PlotLabel -> "Exact (solid)
Asymptotic (dashed)"]


```



The agreement is noticeably better.

Finally we look at the exact and approximate K_0 .

```
compKgraph = Plot[{BesselK[0, x], K0Large[x]}, {x, 0.01, 4},
  PlotRange -> {0, 1.0}, PlotStyle -> {Dashing[0], Dashing[{0.01, 0.01]}},
  AxesLabel -> {"x", "K0(x)"}, PlotLabel -> "Exact (solid)
  Asymptotic (dashed)"]
```



Again we see a good but not great approximation, and again it is possible to develop more accurate approximations.

Very roughly speaking, for large x , $I_0(x)$ behaves like e^x/\sqrt{x} and $K_0(x)$ behaves like e^{-x}/\sqrt{x} . For small x , I_0 is well-behaved, and K_0 becomes logarithmically singular as $x \rightarrow 0$.

5. The Bessel Functions I_m and K_m

The modified Bessel functions $I_m(x)$ and $K_m(x)$ are solutions of equation (5), the modified Bessel's equation of order m . We give here a very brief summary of the basic properties of these functions.

Equation (5) has a regular singular point at $x = 0$, and the Frobenius analysis shows that the roots of the indicial equation are $-m$ and m , and thus they differ by an integer. Further analysis shows that only the root $+m$ gives a solution of the Frobenius form, and this solution, appropriately standardized, is the one called I_m . The series obtained from the Frobenius analysis is

$$I_m(x) = \left(\frac{x}{2}\right)^m \left(\frac{1}{m!} + \frac{(x/2)^2}{(1!)(1+m)!} + \frac{(x/2)^4}{(2!)(2+m)!} + \frac{(x/2)^6}{(3!)(3+m)!} + \dots \right) = \left(\frac{x}{2}\right)^m \sum_{k=0}^{\infty} \frac{(x/2)^{2k}}{(k!)(k+m)!}. \quad (19)$$


The second solution, K_m , has a logarithmic singularity at $x = 0$, as well as a singularity of the form x^{-m} . The series for K_m , which is difficult to derive, is given by

$$K_m(x) = \frac{1}{2} (x/2)^{-m} \sum_{k=0}^{m-1} \frac{(-1)^k (m-k-1)!}{k!} \left(\frac{x}{2}\right)^{2k} + (-1)^{m+1} \ln\left(\frac{1}{2}x\right) I_m(x) + (-1)^m \frac{1}{2} (x/2)^m \sum_{k=0}^{\infty} \{\psi(k+1) + \psi(m+k+1)\} \frac{(x/2)^{2k}}{k!(m+k)!}, \quad (20)$$

where ψ is the Digamma function defined earlier (just after equation (10)).

These functions are both built in. The *Mathematica* function `BesselI[m,x]` gives $I_m(x)$, and the function `BesselK[m,x]` gives $K_m(x)$. Let's see what these functions look like. We first plot I_m for m equal to 0 through 5.


```
graphIm = Plot[{BesselI[0, x], BesselI[1, x], BesselI[2, x], BesselI[3, x],
  BesselI[4, x], BesselI[5, x]}, {x, 0, 4}, AxesLabel -> {"x", "Im"}]
```



We see the apparent exponential growth as before, but we also see that the higher order functions are slower to get started. The reason is that $I_m(x)$ has a factor x^m , which means that the function and the first $m - 1$ derivatives vanish at $x = 0$.

Now let's take a look at a plot of K_m for $m = 0$ through 5. We start with a slight offset in x to avoid the singularity at $x = 0$.

```
graphKm = Plot[{BesselK[0, x], BesselK[1, x], BesselK[2, x], BesselK[3, x], BesselK[4, x],
  BesselK[5, x]}, {x, 0.01, 4}, PlotRange -> {0, 5}, AxesLabel -> {"x", "Km"}]
```



We see the apparent exponential decay for large x as before, and the singularity at $x = 0$, which becomes stronger as m increases.


There are simple asymptotic approximations for $I_m(x)$ and $K_m(x)$ for large x , given by

```
ILarge[m_, x_] := Exp[x] / Sqrt[2 * π * x]
```

```
KLarge[m_, x_] := Exp[-x] * Sqrt[π / (2 * x)]
```

By way of example, we compare the exact and asymptotic expressions for $m = 3$, beginning with I_m .


```
compI3graph = Plot[{BesselI[3, x], ILarge[3, x]},
  {x, 3, 5}, PlotStyle -> {Dashing[0], Dashing[{0.01, 0.01]}},
  AxesLabel -> {"x", "I3(x)"}, PlotLabel -> "Exact (solid)
  Asymptotic (dashed)"]
```



We see that the agreement is good qualitatively, but only fair quantitatively.

Finally, we look at the exact and asymptotic functions for K_3 . We choose the x and y ranges to best illustrate the comparison.

```
compK3graph = Plot[{BesselK[3, x], KLarge[3, x]},
  {x, 0.01, 4}, PlotStyle -> {Dashing[0], Dashing[{0.01, 0.01]}},
  AxesLabel -> {"x", "K3(x)"}, PlotLabel -> "Exact (solid)
  Asymptotic (dashed)"]
```



Again the right qualitative behavior, but only fair quantitative agreement. For both I_m and K_m we can find more accurate multi-term asymptotic approximations for large x .

6. Examples of Bessel Expansions

■ 6.1 Introduction

In this notebook, we construct and plot the partial sums of a Fourier-Bessel series for two specific examples. (For a much more general approach to computing and graphing partial sums, see the notebook entitled Convergence of Bessel Expansions.) In this notebook we use only the functions J_0 . These functions arise as the eigenfunctions of the Sturm-Liouville problem given below.

$$\frac{d}{dr} \left(r \frac{d\psi}{dr} \right) + \lambda r \psi = 0, \quad 0 < r < a, \quad \text{with } \psi \text{ well behaved at } r = 0, \text{ and } \psi(a) = 0.$$

We have $\psi_n(r) = J_0(\alpha_n r/a)$, where α_n is the n th root of J_0 . These eigenfunctions are orthogonal on $[0, a]$ with respect to the weight function r . Any piecewise smooth function $f(r)$ on $[0, a]$ can be expanded in these eigenfunctions. The coefficients in the expansion are calculated using orthogonality. The result is

$$f(r) = \sum_{n=1}^{\infty} C_n \psi_n(r) \quad , \quad \text{where } C_n = \frac{\int_0^a f(r) \psi_n(r) r dr}{\int_0^a \{\psi_n(r)\}^2 r dr}$$

■ 6.2 Example: $f(r) = a^2 - r^2$ on $0 \leq r \leq a$

We now illustrate this theory by expanding the function

```
f[r_] := a^2 - r^2
```

We choose the value 3 for a :

```
a = 3;
```

We find the first 21 zeros of J_0 and assign them to the list named α .

```
 $\alpha$  = N[Table[BesselJZero[0, i], {i, 1, 21}]]
{2.40483, 5.52008, 8.65373, 11.7915, 14.9309, 18.0711,
 21.2116, 24.3525, 27.4935, 30.6346, 33.7758, 36.9171, 40.0584,
 43.1998, 46.3412, 49.4826, 52.6241, 55.7655, 58.907, 62.0485, 65.19}
```

We define the eigenfunctions for *Mathematica*.

```
 $\psi$ [r_, n_] := BesselJ[0, r *  $\alpha$ [n]] / a
```

We call the n th expansion coefficient `coeff[n]`. The theoretical expression is given above in equation (1). We define expressions for both the numerator and the denominator numerically in terms of the *Mathematica* function `NIntegrate`. We also obtain the coefficients below from an analytical expression derived in class. With more complicated functions $f[r]$, numerical integration may be the only choice.

```
num[n_] := NIntegrate[f[r] *  $\psi$ [r, n] * r, {r, 0, a}]
```

```
den[n_] := NIntegrate[( $\psi$ [r, n])^2 * r, {r, 0, a}]
```

We calculate the first 21 coefficients, and assign them to a list named `coeff`.

```
coeff = Module[{ans, c, n}, ans = {};
  Do[{c = num[n] / den[n]; ans = Append[ans, c]}, {n, 1, 21}]; ans]
{9.9722, -1.258, 0.409288, -0.188918, 0.104726, -0.0649906, 0.0435408, -0.0308311,
 0.0227658, -0.0173713, 0.0136099, -0.0108969, 0.00888466, -0.00735654, 0.00617251,
 -0.00523902, 0.00449182, -0.0038857, 0.00338819, -0.00297548, 0.00262987}
```

In class, we also derived an analytical expression for these coefficients. Let's verify that we get the same result that way.

```
analeff[n_] := (8 a^2) / (( $\alpha$ [n])^3 BesselJ[1,  $\alpha$ [n]]);
```

```
analycoefflist =
```

```
Module[{ans, c, n}, ans = {}; Do[{c = analeff[n]; ans = Append[ans, c]}, {n, 1, 21}]; ans];
```

We compare the two results by taking their difference.

```
coeff - analycoefflist
```

```
{-8.88178 × 10-15, -4.44089 × 10-15, -2.38698 × 10-14, 3.43059 × 10-14,
 7.4829 × 10-14, 9.83241 × 10-14, -1.88911 × 10-13, -1.42483 × 10-13, -1.31995 × 10-13,
 -1.13333 × 10-13, 2.02893 × 10-14, -5.02948 × 10-14, -5.39759 × 10-14,
 -2.28427 × 10-13, -9.65981 × 10-15, 3.44247 × 10-14, -3.7346 × 10-14,
 -2.73956 × 10-15, -8.99766 × 10-14, -4.73168 × 10-14, 1.19097 × 10-14}
```

We get excellent agreement. Note that the fourth coefficient is only about 2% of the first coefficient, and the rest are even smaller. This suggests rapid convergence, which we are about to verify with some graphs. This is not so surprising given that the function being expanded satisfies the same zero boundary condition at the edge as the eigenfunctions.

Now we define the k th partial sum of our Fourier-Bessel series.


```
fourbess[r_, k_] := Sum[coeff[n] *  $\psi$ [r, n], {n, 1, k}]
```

Finally, we define `graph[k]`, which produces a graph of $f[r]$ in blue and the k th partial sum of the Fourier-Bessel series in red. We use `pltrange` as a variable containing the plot range. We use `SetOptions` to set the value of the plotting option `ImageSize`. We set it to 250, a good value for printing. For computer visualization, a value of 350 is better.

```
pltrange = {0, 9.2};
SetOptions[Plot, ImageSize -> 250];
graph[k_] := Plot[{f[r], fourbess[r, k]}, {r, 0, a},
  PlotRange -> pltrange, AxesLabel -> {"r", SequenceForm["f[r] = ", f[r]]},
  PlotLabel -> Row[{"k = ", PaddedForm[k, 3]}], PlotStyle ->
  {{RGBColor[0, 0, 1], Thickness[0.004]}, {RGBColor[1, 0, 0], Thickness[0.004]}}
```

Now we look at the first 6 partial sums.

```
GraphicsGrid[Table[{graph[i], graph[i + 1]}, {i, 1, 5, 2}]]
```



The graphs verify our prediction of good convergence.

■ 6.3 Example: $f(r) = 1$ on $0 \leq r \leq a$

Let's do one more example with a different function. We must re-define `f`, and then re-execute the calculation of the coefficients. We choose `f` to be a constant.

```
f[r_] := 1
```

```
coeff = Module[{ans, c}, ans = {};  
  Do[(c = num[n] / den[n]; ans = Append[ans, c]), {n, 1, 21}]; ans]  
{1.60197, -1.0648, 0.851399, -0.729645, 0.648524, -0.589543, 0.54418,  
 -0.507894, 0.478012, -0.452851, 0.431284, -0.412531, 0.396028, -0.38136,  
 0.368208, -0.35633, 0.345532, -0.335659, 0.326587, -0.318213, 0.310451}
```

Again we had an analytic expression for these coefficients in class, so again we use those in a consistency check.


```
analccoeff[n_] := 2 / (α[[n]] BesselJ[1, α[[n]]])  
analycoefflist =  
  Module[{ans, c}, ans = {}]; Do[(c = analcoeff[n]; ans = Append[ans, c]), {n, 1, 21}]; ans];
```

We take the difference to compare the numerical and analytical results.

```
coeff - analycoefflist  
{-1.55431 × 10-15, -4.21885 × 10-15, -5.973 × 10-14, -6.68354 × 10-14,  
 -7.21645 × 10-14, 4.65183 × 10-14, -9.76996 × 10-14, 3.70814 × 10-14,  
 -8.27116 × 10-15, -1.17129 × 10-14, 2.88658 × 10-15, -1.62648 × 10-14,  
 -3.7137 × 10-14, 6.60583 × 10-15, -8.82627 × 10-15, 1.52101 × 10-14, 1.94289 × 10-15,  
 -2.14828 × 10-14, -7.77156 × 10-16, 6.10623 × 10-16, 2.04836 × 10-14}
```


Again we see that the agreement is excellent. We continue our calculations using the numerical coefficients. We don't expect the convergence to be as good here, because the function doesn't vanish at the endpoints. Let's do 21 partial sums for this one. We use the command Manipulate to produce this graph sequence. We first produce the graph sequence, and then hand the finished graphs to Manipulate for display.

```
pltrange = {0, 1.5};  
DynamicModule[{mangraph, k}, Do[mangraph[k] = graph[k], {k, 1, 21, 1}];  
  Manipulate[mangraph[k], {k, 1, 21, 1}]]
```



For visualization in the printed version of the notebook, we construct and display every fourth graph in this sequence.

```
GraphicsGrid[Table[{graph[i], graph[i + 4]}, {i, 1, 17, 8}]]
```



The series is clearly converging, but the struggle is very reminiscent of the Fourier series for a square wave, complete with the Gibbs phenomenon. The resemblance is more than accidental. Remember that the Bessel functions for large argument (i.e., large n in the series) behave like damped trig functions, so the convergence issues are mathematically related to those of the Fourier series.

The above calculation has been very inefficient. In the sequence of 21 partial sums, we re-computed at each step all of the terms that appeared in the preceding graph, plus one new term. It would be far more efficient to save the n th partial sum, and then create the $n+1$ st partial sum by computing only the new term and adding it to the n th partial sum. This is exactly what is done in the notebook *Convergence of Bessel Expansions*. In that notebook we will also see how to send the graph output to a Manipulate panel for convenience in visualization.

7. Who Was Bessel?



Friedrich Wilhelm Bessel
1784 - 1846

Bessel was born in Westphalia (Germany) in 1784. In spite of coming from a poor family, he entered business at the age of 15 with an import-export firm. He was successful with the firm, but his natural intellect and curiosity led him through a sequence of studies including languages, geography, navigation, and astronomy and mathematics. At the age of 20 he published a paper on the orbit of Halley's Comet. The German astronomer Wilhelm Olbers was so impressed with this work that he proposed Bessel as an assistant at the Lilienthal observatory. Thus Bessel was faced early in his adult life with the choice between a business career of relative affluence or a career in science with a much more uncertain financial future. He chose the career in science.

Bessel's achievements are far too numerous to describe in detail, but here are a few brief examples. He studied the earth's size and shape, and in 1841 he deduced from his measurements that the ellipticity of the Earth is $1/299$, a remarkably accurate result. He was a pioneer in precision measurement of stellar positions. His measurements of stellar parallax led to some of the first accurate stellar distance determinations. For example, he determined that the star 61 Cygni is about 10.3 light years from the earth. It is not an exaggeration to say that Bessel's work is one of the foundations of our present knowledge of the scale of our universe.

Bessel's now famous equation arose in his studies of the mutual perturbations of planetary positions. He was one of the first to make a serious effort on the notoriously difficult three-body problem in gravitational theory.

8. References

The literature on Bessel functions is enormous. There is a famous 800-page treatise on Bessel functions first published in 1922 and still very useful:

A Treatise on the Theory of Bessel Functions, by G.N. Watson, Cambridge University Press, 2nd edition, 1962.

The single most useful practical reference for users of Bessel functions is

Handbook of Mathematical Functions, ed. M. Abramowitz and I. A. Stegun, National Bureau of Standards, 1964.

This reference has many useful formulas, graphs and tables for a wide variety of special functions. You can no longer buy a hardbound copy for \$6.50 as you could in 1964, but a relatively inexpensive paperback reprint is still available.

A recent update of the above handbook is

NIST Handbook of Mathematical Functions, ed. Fran W.J. Olver., Daniel W. Lozier, Ronald F. Boisvert, and Charles W. Clark, Cambridge University Press, 2010. The complete version of the NIST handbook is available online at <http://dlmf.nist.gov/>.

The ultimate reference on special functions, including Bessel functions, is

Higher Transcendental Functions, Vol. I, II, and III, The Bateman Manuscript Project, ed. A. Erdelyi, W. Magnus, F. Oberhettinger, and F.C. Tricomi, McGraw-Hill, 1953.

Volume II contains over 100 pages of formulas with Bessel functions.

Almost all texts on advanced calculus or mathematical methods for scientists and engineers have some useful information on Bessel functions. There are hundreds of such books. Here are two by way of example -- the first elementary and the second quite advanced:

Advanced Engineering Mathematics, E. Kreyszig, 7th and earlier editions, John Wiley.

Methods of Theoretical Physics, P.M. Morse and H. Feshbach, McGraw-Hill, 1953.

The historical information given in section 6 on Bessel came from the CD version of the Encyclopedia Britannica, and from a web site at St. Andrews University devoted to the history of mathematics. The picture of Bessel also came from that web site. The web site is called the MacTutor History of Mathematics Archive, and its address is

<http://www-groups.dcs.st-andrews.ac.uk/~history/> .